How Relativistic Time Dilation, Length Contraction, and Frame-Dependent Synchronization Are Related: A Tutorial

A. Richard Zacher

v1—2020-02-09

Abstract: *The special theory of relativity introduces three phenomena that are absent from pre-relativity physics: Frame-dependent synchronization, time dilation, and length contraction. These features are closely related to one another, and all are related to the way in which clocks are synchronized. Although many people, including some physicists, find these features counterintuitive and disturbing, the theory depends on all three for its internal consistency. This paper is a tutorial that uses a thought experiment to illustrate the relationship between these phenomena, employing simple calculations based on the Lorentz transformation.*

Introduction

Everybody (every physicist, at least) knows that there are three weird things that happen to time and space in the special theory of relativity, namely:

- Time dilation—clocks that are moving seem to advance more slowly than identical clocks seen at rest.
- Length contraction—objects measured when they are moving are seen to be shortened in the direction of motion compared with when they are measured at rest.
- Frame-dependent synchronization—events that occur at different places but at the same time in one reference frame will appear to occur at different places *and at different times* in other reference frames; and which of the events occurs first, and by what margin, varies with the motion of the observer.

These three aspects of relativity are direct consequences of the Lorentz transformation, which is central to the theory and to its logical consistency. They are tightly connected with one another, to the extent that if you try to formulate a new theory that gets rid of *some* of these features, you will not have the same agreement with experiment that relativity theory has, and your theory may not behave consistently. Of course, it is conceivable that you might add other features to restore the logical consistency and maintain the known empirical agreements, but so far nobody has done that in a way that convinces many physicists that it is an improvement over Einstein's relativity theory as it is currently understood.

In this notebook I will show through a thought experiment how these three phenomena are mathematically connected, using simple calculations based on special relativity. The calculations are carried out in *Mathematica* to add flexibility to vary the calculations with ease. Some familiarity with *Mathematica* notation and conventions will be useful in understanding the exposition.

The thought experiment

Imagine two fleets of space ships moving through space in opposite directions. Within each fleet all the ships are moving at the same velocity (that is, they are all at rest relative to one another), distributed along a line with constant equal spacing δx between the ships. The lines along which the fleets are moving are parallel, and offset just enough to prevent the ships of the two fleets from colliding. The *relative speed* between the two fleets is **v**, with fleet **A** moving in the positive **x** direction and fleet **B** moving in the negative **x** direction. Initially the fleets are separated in the **x** direction, but they are approaching one another at constant relative speed **v**.

There are **n** ships in each fleet, where **n>1**. We will number the ships in each fleet **0,1,...,n-1**, starting with the lead ship in the fleet.

At the precise moment when the ship **0** of either fleet meets a ship of the other fleet, the ship *of the other fleet* flashes a light signal that can be seen (eventually) by all the other ships. Furthermore, after emitting this first flash upon meeting ship **0** of the other fleet, each ship continues to emit flashes at intervals **δt = δx / v** as timed on its own clock. This is the time interval between encounters of any particular ship in fleet **A** with successive ships in fleet **B**, as measured in fleet **B**; it is also the time interval between encounters of any particular ship in fleet **B** with successive ships in fleet **A**, as measured in fleet **A**. (Be sure you read that statement carefully and understand what it means, because if you get it wrong, the whole analysis won't make sense; and bear in mind that this is special relativity, and it makes a difference *whose* time interval measurements we are talking about!).

We specify that the event of the two light flashes when ship **0** of the **A** fleet meets ship **0** of the **B** fleet occurs at **t=0** and **x=0** in both reference frames. The two flashes take place at (almost) the same point in space and at the same time, so observers in both frames will agree that they are simultaneous. Ship **0** in each fleet will therefore emit a flash at time **t=0** in its own frame and at intervals of **δt** thereafter, so these flashes emitted by ship **0** will occur at times **t=i δt**, **i=0,1,2...** in that ship's own frame. Likewise, ship **j** in each fleet will emit flashes at times **t=i δt**, **i=j,j+1,...** in that ship's own frame.

Mathematica preliminaries & tools setup

Here we set up a few definitions that will useful.

```
In[•]:= Clear[δt, δx, v, γ, c, interval]
```

We note that the travel time of ship **0** of fleet **B** from one ship of fleet **A** to the next (a distance of **δx** away as measured in fleet **A**) is **v δt** as measured in fleet **A**; so of course **δt=δx/v**.

```
In[•]:= δx = v δt
Out[•]= v δt
```

It is convenient (and conventional) to define the time dilation constant **γ[v]** in the way given by the following substitutions:

```
In[•]:= γ[-v_] := γ[v]
```

In[•]:= $\gamma\text{Expand} = \left\{\gamma[\text{v\_}] \to \frac{1}{\sqrt{1 - \frac{v^2}{c^2}}}\right\};$

In[•]:= $\gamma\text{Sub} = \left\{v^2 \to c^2 \left(1 - \frac{1}{\gamma[v]^2}\right)\right\};$

We will be working in two-dimensional Minkowski space, expressing the spacetime locations of events by vectors like **{t, x}**. A displacement vector **d12** between two events **{t1, x1}** and **{t2, x2}** can then be calculated easily:

```
In[•]:= d12 = {t2, x2} - {t1, x1}
Out[•]= {-t1 + t2, -x1 + x2}
```

An important feature of Minkowski space is that, no matter what basis is used (that is, no matter what frame of reference you are examining events from), there is a quantity called the *invariant interval* associated with any pair of events that has the same value regardless of the frame of reference. This interval is the inner product of the spacetime displacement vector between the two events, using the Minkowski metric; it is defined for any displacement vector as follows:

In[•]:= $\text{interval}[\{\Delta t\_, \Delta x\_\}] := c^2\, \Delta t^2 - \Delta x^2$

so that for the displacement **d12** computed above, we have the interval

```
In[●]:= interval[d12]
```

$$\text{Out}[\bullet]= c^2 (-t1 + t2)^2 - (-x1 + x2)^2$$

Setting up the example and doing the calculation

First we describe the coordinate pairs **{t,x}** of the flash events as recorded in the frames of the ships that produce them. The identifier notation for the coordinate pairs is **xAA[i,j]**, **xAB[i,j]**, **xBB[i,j]**, and **xBA[i,j]**, where, for example, **xAB[i,j]** is the *coordinate pair measured in the frame of* fleet **B** (second letter after the **x**) for the flash emitted by the $\mathbf{j}^{\text{th}}$ ship of fleet **A** (first letter after the **x**) at time **i δt** as timed in the frame of the fleet **A**. Likewise, **xAA[4,2]** would be the *coordinate pair measured in the frame of* fleet **A** for the flash occurring at ship **2** (the third ship!) of fleet **A**, at time **4 δt** after time **0** as timed in the reference frame of fleet **A**.

As we have described the setup above, the formulas that follow are valid for **i ≥ j** and **0 ≤ j ≤ n - 1.**

Here are the formulas for the spacetime coordinate pairs **{t,x}** of the flashes as observed in the *same* fleet in which they occur:

```
In[●]:= xAA[i_, j_] = {i δt, -j δx}
```

$$\text{Out}[\bullet]= \{i\, \delta t, -j\, v\, \delta t\}$$

```
In[●]:= xBB[i_, j_] = {i δt, j δx}
```

$$\text{Out}[\bullet]= \{i\, \delta t, j\, v\, \delta t\}$$

Note that in fleet **A** the ships are numbered from right to left, while in fleet **B** they are numbered from left to right—starting at zero for the leading ship in both instances.

To get the coordinates of the flash events as recorded in the frame of the other fleet, we will have to apply the appropriate Lorentz transformation, which I'll call **lorentzAB[v]** for a transformation *from* coordinates measured at rest in fleet **A** *to* coordinates measured at rest in fleet **B**, where fleet **A** is moving at velocity **+v** relative to fleet **B**.

$$\text{In}[\bullet]:= \text{lorentzAB}[v\_] = \gamma[v] \left\{\left\{1, \frac{v}{c^2}\right\}, \{v, 1\}\right\};$$

```
In[●]:= lorentzAB[v] // MatrixForm
```

$$\text{Out}[\bullet]\text{//MatrixForm}= \begin{pmatrix} \gamma[v] & \frac{v\,\gamma[v]}{c^2} \\ v\,\gamma[v] & \gamma[v] \end{pmatrix}$$

```
In[●]:= lorentzBA[v_] = lorentzAB[-v];
```

```
In[●]:= lorentzBA[v] // MatrixForm
```

$$\text{Out}[\bullet]\text{//MatrixForm}= \begin{pmatrix} \gamma[v] & -\frac{v\,\gamma[v]}{c^2} \\ -v\,\gamma[v] & \gamma[v] \end{pmatrix}$$

```
In[●]:= lorentzAB[v].{t, x}
```

$$\text{Out}[\bullet]= \left\{t\,\gamma[v] + \frac{v\,x\,\gamma[v]}{c^2}, t\,v\,\gamma[v] + x\,\gamma[v]\right\}$$

We have, then,

In[∘]:= `xAB[i_, j_] = ((lorentzAB [v].xAA[i, j] //. γSub) // Simplify)`

Out[∘]= $\left\{\frac{j\,\delta t}{\gamma[v]} + (i - j)\,\delta t\,\gamma[v],\ (i - j)\,v\,\delta t\,\gamma[v]\right\}$

In[∘]:= `xBA[i_, j_] = ((lorentzBA[v].xBB[i, j] //. γSub) // Simplify)`

Out[∘]= $\left\{\frac{j\,\delta t}{\gamma[v]} + (i - j)\,\delta t\,\gamma[v],\ (-i + j)\,v\,\delta t\,\gamma[v]\right\}$

If we are just interested in time coordinates of events, we can extract them as follows:

In[∘]:= `tAA[i_, j_] = First[xAA[i, j]]`

Out[∘]= $i\,\delta t$

In[∘]:= `tAB[i_, j_] = First[xAB[i, j]]`

Out[∘]= $\frac{j\,\delta t}{\gamma[v]} + (i - j)\,\delta t\,\gamma[v]$

In[∘]:= `tBA[i_, j_] = (First[xBA[i, j]])`

Out[∘]= $\frac{j\,\delta t}{\gamma[v]} + (i - j)\,\delta t\,\gamma[v]$

In[∘]:= `tBB[i_, j_] = First[xBB[i, j]]`

Out[∘]= $i\,\delta t$

Now let’s look at the relative timing of successive flashes from the same ship. These flashes occur at intervals of **δt=δx/v** as seen by other ships in the same fleet, while the ship hasn’t moved relative to the other ships in its fleet:

In[∘]:= `tAA[i + 1, j] - tAA[i, j] // Simplify`

Out[∘]= δt

In[∘]:= `tAB[i + 1, j] - tAB[i, j] // Simplify`

Out[∘]= $\delta t\,\gamma[v]$

The timing between flashes from each ship has been set in a peculiar way. The flashes from ship **0** of fleet **A** occur at a spacing **δt**, which is the time it takes for ship **0** of fleet **B** to move from one fleet **A** ship to the next, as seen in fleet **A**. However, it is *not* the time it takes for ship **0** of fleet **A** to move from one ship of fleet **B** to the next, as seen in fleet **A**. This latter time is

In[∘]:= `(((tBA[j + 1, j + 1] - tBA[j, j] //. γExpand) // Simplify) //. γSub) // PowerExpand`

Out[∘]= $\frac{\delta t}{\gamma[v]}$

This time is in fact

In[∘]:= $\frac{\delta x / \gamma[v]}{v}$

Out[∘]= $\frac{\delta t}{\gamma[v]}$

where **δx/γ[v]** is the *Lorentz-contracted distance* between the ships of fleet **B** as seen by fleet **A**.

As seen by the other fleet, the successive flashes from the same ship are farther apart in time by a factor of **γ**, and they are separated in space—not by **δx=v δt**, but by **v γ δt** —the distance the ship is seen to move between flashes that are **γ δt** apart. (Note that **γ[-v]==γ[v]**).

In[•]:= `tAB[i + 1, j] - tAB[i, j] // Simplify`

Out[•]= $\delta t\,\gamma[v]$

In[•]:= `tBA[i + 1, j] - tBA[i, j] // Simplify`

Out[•]= $\delta t\,\gamma[v]$

The above result is what we have learned to expect—it is the result of time dilation in a moving frame. But now let's compare times in another way. We'll compare successive flashes occurring in two successive ships of fleet **A**. The spacetime displacement **dAA** between the flashes, as seen in fleet **A**, is

In[•]:= `dAA = ((xAA[i + 1, j + 1] - xAA[i, j]) //. γExpand // Simplify) //. γSub // PowerExpand`

Out[•]= $\{\delta t, -v\,\delta t\}$

As seen in fleet **A**, these two flashes are exactly **δt** apart in time, but because the second flash occurs in the ship behind the ship emitting the first flash, they occur at a spatial separation of **-v δt** (that is, ship **j+1** in fleet **A** is a distance **δx** to the left of ship **j** as seen by fleet **A**).

Looking at these same two events in fleet **A** from the point of view of fleet **B**, we get a displacement

In[•]:= `dAB = ((xAB[i + 1, j + 1] - xAB[i, j]) //. γExpand // Simplify) //. γSub // PowerExpand`

Out[•]= $\left\{\frac{\delta t}{\gamma[v]}, 0\right\}$

In this case, by the time the **(j + 1)th** ship in fleet **A** emits the flash at time **(i+1) δt** in its own frame, it has reached the same spatial location in the fleet **B** frame that ship **j** in fleet **A** occupied the previous time it emitted a flash. That is why the difference in the spatial locations between these two flashes, as seen in fleet **B**, is **0**. However, note that while the flashes occur at a time difference of **δt** as measured in the frame of fleet **A**, they occur at a time difference of **δt/γ** as measured in fleet **B**. So in this case, the time interval between these events in fleet **A** is not *dilated* when viewed in fleet **B**—it is actually *contracted*! That is, these two flashes that occur at an interval **δt** in fleet **A** are observed to occur at the *shorter* interval **δt/γ** as measured by fleet **B**.

By symmetry, the same thing must happen in reverse, when fleets **B** and **A** look at flashes occurring in succession in two successive ships in fleet **B**:

In[•]:= `dBB = ((xBB[i + 1, j + 1] - xBB[i, j]) //. γExpand // Simplify) //. γSub // PowerExpand`

Out[•]= $\{\delta t, v\,\delta t\}$

In[•]:= `dBA = ((xBA[i + 1, j + 1] - xBA[i, j]) //. γExpand // Simplify) //. γSub // PowerExpand`

Out[•]= $\left\{\frac{\delta t}{\gamma[v]}, 0\right\}$

It is interesting—and important—to note that here we are comparing readings of the two systems of clocks at two different spatial points and two different times in one system, but at a single point and two different times in the other system; and it is the time interval measured in the system where the events are compared *at the same point in space* that has the smaller value. The clocks are moving in the same way that they were moving before, but now we are reading different clocks that are supposedly synchronized.

We can get more insight into what is going on if we look at the invariant interval for the displacement between the flash events as expressed in the basis of each fleet. The displacements between flashes occurring at an interval **δt** apart but on two successive ships in fleet **B** is given in fleet **B** as

```
In[•]:= dBB
Out[•]= {δt, v δt}
```

and in fleet **A** as

```
In[•]:= dBA
```

$$\text{Out[•]=}\ \left\{\frac{\delta t}{\gamma[v]},\ 0\right\}$$

From these we compute the interval in the basis of fleet **B**:

```
In[•]:= intervalBB = interval[dBB] // Simplify
```

$$\text{Out[•]=}\ \left(c^2 - v^2\right)\ \delta t^2$$

and in the basis of fleet **A**:

```
In[•]:= intervalBA = interval[dBA] /. γExpand // Simplify
```

$$\text{Out[•]=}\ \left(c^2 - v^2\right)\ \delta t^2$$

As it must be, the interval is independent of the frame of reference in which the events are viewed. However, we can rewrite the time component of the displacement in each frame in terms of the invariant interval and the spatial component of the displacement in that frame. Because the invariant interval is given by $\textbf{interval} == c^2\, dt^2 - dx^2$,
we can compute the magnitude of the time component of displacement between the light flashes as

$$\text{In[•]:=}\ dt = \frac{1}{c}\sqrt{\text{interval} + dx^2}$$

$$\text{Out[•]=}\ \frac{\sqrt{dx^2 + \text{interval}}}{c}$$

Because the *interval* is the same regardless of the frame of reference, and is always positive for timelike displacements (for example, any displacement between events that occur at the same spatial location in *some* inertial frame), it must be that the *time component* **dt** has a larger magnitude in a frame where the magnitude of the spatial component **dx** is larger. This means in particular that the time component of the displacement between two timelike-separated events is a minimum in a frame in which the two events occur at the same spatial location. When applied to single clocks, this means that stationary clocks appear to tick more slowly than moving clocks.

To review:

When we compare two successive light flashes from the same ship, the time between the flashes as seen on that ship (that is, at the same spatial location) is δt; but as observed from the other fleet, the time between those same flashes, which occur at different locations $\gamma\, v\, \delta t$ apart, is $\gamma\, \delta t$.

On the other hand, when we compare successive light flashes from different ships in the same fleet at spatial locations $v\delta t$ apart, the flashes are δt apart as timed in the same fleet. As observed in the other fleet, those two flashes occur at the same location, but separated by a time interval $\delta t/\gamma$.

So apparently it isn't that the clocks are running slower in either frame; rather, the time dilation effect results from the fact that clocks at different spatial points are synchronized differently in the two frames. In both situations, it is when the two events being timed occur at the same spatial point in one frame that the time difference for those events in that frame is less by a factor of $1/\gamma$ than the time between the same events observed in the other frame.

What do we learn from this?

The lesson here is, I think, that time dilation is *not* simply a slowing down of moving clocks. It’s subtler than that. It’s more accurate to think of it as a phenomenon closely connected with clock synchronization.

Time and space are coupled in relativity theory. It’s not enough to say that time slows down in a moving frame. It can appear to speed up, too, as we have just seen. It depends on where the events are that you are timing.

It is both nonsense and a gross oversimplification to say that “Your clocks run slower than my clocks, and my clocks run slower than your clocks.” Which clocks appear to run slower depends on how you look at them, and especially where they are when you look at them.

In summary, I’ll emphasize the key point that I am trying to make: If you are going to replace relativity theory as a fundamental physical theory, it won’t do to replace parts of it piecemeal. If you decide to keep time dilation while throwing out frame-dependent synchronization and length contraction, you will likely come up with a muddle that doesn’t hang together well. That is not to say that you could not possibly come up with an alternative theory that works. Just don’t expect it to be easy!